
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of ESM test case 10.8.1 which are part of the LTE test suite.

The test case can be demonstrated to run with one LTE UE (see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Cases:
tc_10_8_1
Test Group:
ESM

ATS Version:
iwd-EUTRA-TVB2009-12_D10wk33
System Simulator used:
Anite Conformance Toolset
UE(s) used:                       LG L2000 UE

Verification Status:
PASS
4 Corrections required for test case 10.8.1

4.1 Introduction

This section describes the changes required to make test case 10.8.1 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-TVB2009-03_D10wk33 release.

4.2 Change 1

	Testcase name
	function f_TC_10_8_1_EUTRA()

	Reason for change
	1) There are two EPS dedicated bearers in this test case so need to set up two additional DRBs before preamble.
2) As per 36.523-1, preamble requires establishing default and dedicated EPS bearers.

	Summary of change
	1) The parameter value to setup additional DRBs in function call f_EUTRA_CellConfig_DRB_Common is changed to 2 from 1 before preamble.

2) Function f_EUTRA_RbEst_Def replaced with f_EUTRA_RbEst_508_AM which establishes dedicated EPS bearer 6 as well.



	Source of change
	File : …\10\ESM_testcases.ttcn


Before:
	function f_TC_10_8_1_EUTRA() runs on EUTRA_PTC

  {

    f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, LTEsinglecell);

    // Create cell and send out system information

    f_EUTRA_CellConfig_DRB_Common(eutra_CellA, 1, 0);
    f_EUTRA_Preamble(eutra_CellA, STATE2_IDLEUPDATE);

    f_EUTRA_RbEst_Def(eutra_CellA);
    f_EUTRA_TestBody_Set(true);

    fl_TC_10_8_1_Body();

    // Postamble

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_CellA, E2_CONNECTED);

  } // function f_TC_10_8_1_EUTRA


After:
	 function f_TC_10_8_1_EUTRA() runs on EUTRA_PTC

  {

      f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, LTEsinglecell);

    // Create cell and send out system information

    f_EUTRA_CellConfig_DRB_Common(eutra_CellA, 2, 0);
    f_EUTRA_Preamble(eutra_CellA, STATE2_IDLEUPDATE);

    f_EUTRA_RbEst_508_AM(eutra_CellA);  

    f_EUTRA_TestBody_Set(true);

    fl_TC_10_8_1_Body();

    // Postamble

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_CellA, E2_CONNECTED);

  } // function f_TC_10_8_1_EUTRA


4.3 Change 2

	Testcase name
	function fl_TC_10_8_1_Body()

	Reason for change
	1) To trigger Bearer Resource Modification Request on UE, modified QoS and TFT in should be provided by AT commands at step 1.

2) A) As per Step 1 in 36.523-1 Table 10.8.1.3.2-1, UE should trigger Bearer Resource Modification Request on dedicated EPS bearer 6.

B) The TFT operation code in TFT IE in Bearer Resource Modification Request should be “Replace packet filters in existing TFT” instead of “Delete packet filters from existing TFT” in step 2.

C) The number of packet filters expected in TFT IE should be 1 since the AT command is used to modify only one filter ID  parameters in step 2 according to  reference dedicated EPS bearer context#2 in table 6.6.2-1 TS 36.508

3) In step 3

I. DRB2 is already setup with an EPS dedicated bearer 6

II. EPS dedicated bearer ID 6 is already used in preamble.

III. QCI as per reference dedicated EPS bearer context#2 can not be used since as per 23.401v880 5.4.2.1 modification from a QCI of resource type non-GBR to a QCI of resource type GBR and vice versa is not possible.

IV. The TFT template should be included for dedicated EPS bearer 7 not for dedicated EPS bearer 6.

4) Since new EPS dedicated bearer 7 is allocated in step 3. So, the dedicated EPS bearer expected in step 4 should be 7.

Note: Prose CR R5-104168 is agreed at RAN5 meeting.

	Summary of change
	1) New template cs_508_Tft_Dedicated_1_Modify and f_Get_508_PacketFilter1_Modify is added to trigger DL filter ID 0 on EPS dedicated bearer 6 to be modified as bidirectional Filter as per “reference dedicated EPS bearer context#2” in table 6.6.2-1 TS 36.508. Modified function call to function f_UT_RequestModifyDedicatedBearer to pass default Bearer ID, TFT and QoS parameters. (Refer change 3)
2) A) New template cr_508_BearerResourceModReq_10_8_1 is created in which the EPS bearer id 6 set as an IE value for epsBearerIdentityforPacketFilter IE in Bearer Resource Modification Request message.

New template cr_Tft_10_8_1 is created in which TFT IE operation code is changed to “Replace packet filters in existing TFT”  and the number of packet filters is changed to 1 since the AT command used to modify TFT parameters for one filter ID only according to  reference dedicated EPS bearer context#2 in table 6.6.2-1 TS 36.508
3) In step 3

I. DRB3 setup instead of DRB2

II. EPS dedicated bearer ID 7 is allocated

III. QCI as per reference dedicated EPS bearer context#1 and the same QCI which is received in Bearer Resource Modification Request is sent in step 3 with IEI value as ‘omit’ since this IE is mandatory in this step.

IV. TFT template changed as per “reference dedicated EPS bearer context#2” in table 6.6.2-1 TS 36.508.

4) The expected EPS dedicated bearer is changed to 7 in Activate Dedicated EPS Bearer Context Accept message in step 4.

	Source of change
	File : …\10\ESM_testcases.ttcn


Before:
	function fl_TC_10_8_1_Body() runs on EUTRA_PTC

  {

    var ProcedureTransactionIdentifier v_EpsPti := '01'O;

    var SRB_COMMON_IND v_ReceivedAsp;

    // Step 1. Cause the UE to request bearer resource modification of

    //          dedicated EPS bearer associated with first PDN connectivity.

    f_UT_RequestModifyDedicatedBearer (UT,

                                       hex2int(tsc_EpsDedicatedBearerId));
    // Step 2. Check: Does the UE transmit a BEARER RESOURCE MODIFICATION REQUEST message? (1,P)

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_BearerResourceModReq(?,

                                                                                  tsc_EpsDefaultBearerId,
                                                                                  omit)))) -> value v_ReceivedAsp;

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    // Store procedure transaction id to be used in SS responses.

    v_EpsPti :=  v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.bEARER_RESOURCE_MODIFICATION_REQUEST.procedureTransactionIdentifier;

    // Step 3. The SS transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST message.

    f_EUTRA_TxDedicatedEpsBearerCtxtReq(eutra_CellA,

                                        tsc_DRB2,
                                        tsc_EpsDedicatedBearerId,
                                        v_EpsPti,

                                        cs_LinkedId(hex2bit( tsc_EpsDefaultBearerId)),

                                        cds_EPS_QoS_Dedi_nonGBR_QCI,

                                        cs_508_Tft_Dedicated_2(tsc_EpsDedicatedBearerId));
    // Step 4. Check: Does the UE transmit an ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT message? (2,P)

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActDedicatedEPSBearerCxtAccept(tsc_EpsDedicatedBearerId))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 4");

  } // function fl_TC_10_8_1_Body


After:
	function fl_TC_10_8_1_Body() runs on EUTRA_PTC

  {

    var ProcedureTransactionIdentifier v_EpsPti := '01'O;

    var EPS_BearerIdentity v_DedicatedEpsBearerId2 := '7'H;

    var EPS_QualityOfService v_Qos;
    var SRB_COMMON_IND v_ReceivedAsp;

    // Step 1. Cause the UE to request bearer resource modification of

    //          dedicated EPS bearer associated with first PDN connectivity.

    f_UT_RequestModifyDedicatedBearer (UT,

                                       hex2int(tsc_EpsDedicatedBearerId),

                                       hex2int(tsc_EpsDefaultBearerId),

                                       cs_508_EPS_QoS_Dedicated_1, //Can  n't be changed from GBR to NonGBR and vice-versa

                                       cs_508_Tft_Dedicated_1_Modify (tsc_EpsDedicatedBearerId));
    // Step 2. Check: Does the UE transmit a BEARER RESOURCE MODIFICATION REQUEST message? (1,P)

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_BearerResourceModReq_10_8_1(?,

                                                                                  tsc_EpsDedicatedBearerId,

                                                                                  omit)))) -> value v_ReceivedAsp;

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    // Store procedure transaction id to be used in SS responses.

    v_EpsPti :=  v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.bEARER_RESOURCE_MODIFICATION_REQUEST.procedureTransactionIdentifier;

    // Step 3. The SS transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST message.

    //Following line modified as per 36.523-1 Table 10.8.1.3.3-1

     v_Qos := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.bEARER_RESOURCE_MODIFICATION_REQUEST.requiredTrafficFlowQos;

    v_Qos.iei := omit;
    f_EUTRA_TxDedicatedEpsBearerCtxtReq(eutra_CellA,

                                        tsc_DRB3,
                                        v_DedicatedEpsBearerId2,
                                        v_EpsPti,

                                        cs_LinkedId(hex2bit( tsc_EpsDefaultBearerId)),

                                        v_Qos,

                                        cs_508_Tft_Dedicated_2(tsc_EpsDedicatedBearerId2));
    // Step 4. Check: Does the UE transmit an ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT message? (2,P)

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActDedicatedEPSBearerCxtAccept(v_DedicatedEpsBearerId2))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 4");

  } // function fl_TC_10_8_1_Body


New Template:

	template NAS_UL_Message_Type cr_508_BearerResourceModReq_10_8_1(template (present) ProcedureTransactionIdentifier p_ProcTi,

                                                         EPS_BearerIdentity p_LinkedEpsBearerId,

                                                         template (omit) ESM_Cause p_Cause) :=

{

  bEARER_RESOURCE_MODIFICATION_REQUEST := {

    epsBearerId                    := '0'H,

    protocolDiscriminator          := tsc_PD_ESM,

    procedureTransactionIdentifier := p_ProcTi,

    messageType                    := tsc_MT_BearerResourceModificationRequest,

    spareHalfOctet                 := tsc_SpareHalfOctet,

    epsBearerIdentityforPacketFilter := cs_LinkedId(hex2bit(p_LinkedEpsBearerId)),   // of the associated dedicated bearer

    trafficFlowAgg                 := cr_Tft_10_8_1('100'B),  // Replace packet filters from existing TFT

    requiredTrafficFlowQos         := cdr_EPS_QoS_Any_iei,

    esmCause                       := p_Cause,

    protocolConfigurationOptions   := cr_ProtocolConfigOptionsAny ifpresent

  }

}; 

template TrafficFlowTemplate cr_Tft_10_8_1 (B3_Type p_Code) :=

{ /* 24.301 cl. 9.9.4.13 -> 24.008 cl. 10.5.6.12 */

  iei := omit, /* present in case of TLV; omit in case of LV */

  iel := ?,

  tftOperationCode := p_Code,

  eBit := ?,

  noOfPktFilter := '0001'B, // value must be greater than 0 and <= 16 (only 4 bits)

  packetFilterList := ?,

  parameterList := *

};

template (value) TrafficFlowTemplate cs_508_Tft_Dedicated_1_Modify (HalfOctet_Type p_EPS_BearerId := tsc_EpsDedicatedBearerId):=

{ // According to TS 36.508 clause 6.6.2 context#1

  /* @status    APPROVED */

  /* @sic R5-100771 sic@ */

  iei   := omit, //  '36'O,

  iel   := '0D'O,

  tftOperationCode := '001'B, // Create new TFT

  eBit  := '0'B, // Parameter list not included

  noOfPktFilter := '0001'B,

  packetFilterList := {f_Get_508_PacketFilter1_Modify(1, p_EPS_BearerId)},

  parameterList := omit

};

function f_Get_508_PacketFilter1_Modify(integer p_PacketFilterNumber,

                                  HalfOctet_Type p_EPS_BearerId) return template (value) PacketFilter

  {

    var integer v_FilterCount := 0;

    var template (value) PacketFilter v_Filter;

    if (pc_IPv4){

      v_Filter.contents[0] := cs_PktFilterCompIPv4RemoteAddress;

      v_FilterCount := v_FilterCount + 9;

    }

    else { // IPv6

      v_Filter.contents[0] := cs_PktFilterCompIPv6RemoteAddress;

      v_FilterCount := v_FilterCount + 33;

    }

    select (p_PacketFilterNumber){

      case (1) {

        v_Filter.iei         := '30'O;

        v_Filter.precedence  := '00001111'B;

      }

    }

    v_Filter.iel := int2oct(v_FilterCount, 1);

    return v_Filter;

  }




4.4 Change 3

	Testcase name
	10.8.1

	Reason for change
	New TFT and QCI parameters need to be provided for AT commands to trigger Bearer Resource Modification Request from UE in step 1.



	Summary of change
	Function f_UT_RequestModifyDedicatedBearer modified to provide new TFT and QCI parameters with AT commands to UE in step 1.


	Source of change
	File : …\Common\UpperTesterFunctions.ttcn


Before:
	function f_UT_RequestModifyDedicatedBearer (UT_PTC_MTC_PORT p_Port,

                                              integer p_DedicatedId)

  {

    fl_UT_ApplyCommand(p_Port, cas_UT_Req(MODIFY_BEARER, NO_CNF_REQUIRED, cs_ModifyPDP (int2str(p_DedicatedId))));

  };


After:
	function f_UT_RequestModifyDedicatedBearer (UT_PTC_MTC_PORT p_Port,

                                              integer p_DedicatedId,

                                              integer p_DefaultId,

                                              template (value) EPS_QualityOfService p_QoS,

                                              template (value) TrafficFlowTemplate p_Tft)

  {

    var EPS_QualityOfService v_QoS := valueof (p_QoS); // need to do this to then manipulate it in the charstring

    var TrafficFlowTemplate v_Tft := valueof (p_Tft); // need to do this to then manipulate it in the charstring

    var integer v_I, v_NoOfPacketFilters := bit2int (v_Tft.noOfPktFilter);

    var charstring v_QoSstring := "";

    v_QoSstring := int2str(p_DedicatedId);

    v_QoSstring := v_QoSstring &","& int2str(bit2int(v_QoS.qci));

    v_QoSstring := v_QoSstring &","& int2str(bit2int(v_QoS.guaranteedBitRateDl));

    v_QoSstring := v_QoSstring &","& int2str(bit2int(v_QoS.guaranteedBitRateUl));

    v_QoSstring := v_QoSstring &","& int2str(bit2int(v_QoS.maxBitRateDl));

    v_QoSstring := v_QoSstring &","& int2str(bit2int(v_QoS.maxBitRateUl));

    // Extended rates are not dealt with at the moment (all set to 0 in current 36.508) FFS

    fl_UT_ApplyCommand(p_Port, cas_UT_Req(REQUEST_DED_BEARER, NO_CNF_REQUIRED, cs_SecondaryPDP(int2str(p_DedicatedId), int2str(p_DefaultId))));

    fl_UT_ApplyCommand(p_Port, cas_UT_Req(DEFINE_EPS_QOS, NO_CNF_REQUIRED, cs_EPSQoS(v_QoSstring)));

    if (v_NoOfPacketFilters != 0) {

      for (v_I := 0; v_I < v_NoOfPacketFilters; v_I := v_I + 1)

        {

          fl_UT_ApplyCommand(p_Port, cas_UT_Req(DEFINE_TFT, NO_CNF_REQUIRED, cs_TFT(fl_GenerateATForTFTPacketFilter (p_DedicatedId, v_Tft.packetFilterList[v_I]))));

        }

    }
    fl_UT_ApplyCommand(p_Port, cas_UT_Req(MODIFY_BEARER, NO_CNF_REQUIRED, cs_ActivatePDP(int2str(p_DedicatedId))));

 };


4.5 Change 4

	Testcase name
	cs_AT_MODIFYBEARER  

	Reason for change
	The extra comma preceding before EPS bearer ID in CGCMOD AT command is not correct syntax

	Summary of change
	The extra comma is removed from CGCMOD AT command

	Source of change
	File : …\MTC\MTC_UpperTester.ttcn


Before:
	template (value) Ut_System_Cmd_Type cs_AT_MODIFYBEARER  (Ut_ParameterList_Type p_ParamList) :=

    cs_AT_Cmd("AT+CGCMOD=,"& p_ParamList[0].Value &"<CR>");          /* @status    APPROVED */


After:
	template (value) Ut_System_Cmd_Type cs_AT_MODIFYBEARER  (Ut_ParameterList_Type p_ParamList) :=

    cs_AT_Cmd("AT+CGCMOD="& p_ParamList[0].Value &"<CR>");      /* @status    APPROVED */


4.6 Change 5

	Testcase name
	function f_Get_508_PacketFilter

	Reason for change
	Same filter ID and precedence used for both uplink and downlink in a single TFT

Note: Prose CR R5-104702 is agreed for this at RAN5 meeting.

	Summary of change
	Changes done as per CR R5-104702

	Source of change
	File : …\Common_NAS\NAS_AuxiliaryDefsAndFunctions.ttcn


Before:
	function f_Get_508_PacketFilter(integer p_PacketFilterNumber,

                                  HalfOctet_Type p_EPS_BearerId) return template (value) PacketFilter

  {

    var integer v_FilterCount := 0;

    var template (value) PacketFilter v_Filter;

    if (pc_IPv4){

      v_Filter.contents[0] := cs_PktFilterCompIPv4RemoteAddress;

      v_FilterCount := v_FilterCount + 9;

    }

    else { // IPv6

      v_Filter.contents[0] := cs_PktFilterCompIPv6RemoteAddress;

      v_FilterCount := v_FilterCount + 33;

    }

    select (p_PacketFilterNumber){

      case (1) {

        v_Filter.iei         := '10'O;

        v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6), 8);

        v_Filter.contents[1] := cs_PktFilterCompSingleRemotePort31160(p_EPS_BearerId);

        v_FilterCount := v_FilterCount + 3;

        v_Filter.contents[2] := cs_PktFilterCompProtocolIdNextHeader;

        v_FilterCount := v_FilterCount + 2;

      }

      case (2) {

        // Update to values in Reference packet filter #2

        v_Filter.iei         :=  '20'O; // UL only filter, ID=0

        v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6), 8);
        v_Filter.contents[1] := cs_PktFilterCompSingleRemotePort61000(p_EPS_BearerId);

        v_FilterCount := v_FilterCount + 3;

        v_Filter.contents[2] := cs_PktFilterCompProtocolIdNextHeader;

        v_FilterCount := v_FilterCount + 2

          }

      case (3){

        // Update to values in Reference packet filter #3

        v_Filter.iei         :=  '31'O; // Bidirectional filter, ID=1
        v_Filter.precedence  := '00001111'B;

      }

    }

    v_Filter.iel := int2oct(v_FilterCount, 1);

    return v_Filter;

  }


After:
	function f_Get_508_PacketFilter(integer p_PacketFilterNumber,

                                  HalfOctet_Type p_EPS_BearerId) return template (value) PacketFilter

  {

    var integer v_FilterCount := 0;

    var template (value) PacketFilter v_Filter;

    if (pc_IPv4){

      v_Filter.contents[0] := cs_PktFilterCompIPv4RemoteAddress;

      v_FilterCount := v_FilterCount + 9;

    }

    else { // IPv6

      v_Filter.contents[0] := cs_PktFilterCompIPv6RemoteAddress;

      v_FilterCount := v_FilterCount + 33;

    }

    select (p_PacketFilterNumber){

      case (1) {

        v_Filter.iei         := '10'O;

        v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6), 8);

        v_Filter.contents[1] := cs_PktFilterCompSingleRemotePort31160(p_EPS_BearerId);

        v_FilterCount := v_FilterCount + 3;

        v_Filter.contents[2] := cs_PktFilterCompProtocolIdNextHeader;

        v_FilterCount := v_FilterCount + 2;

      }

      case (2) {

        // Update to values in Reference packet filter #2

        v_Filter.iei         :=  '21'O; // UL only filter, ID=0

        v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)+2), 8);
        v_Filter.contents[1] := cs_PktFilterCompSingleRemotePort61000(p_EPS_BearerId);

        v_FilterCount := v_FilterCount + 3;

        v_Filter.contents[2] := cs_PktFilterCompProtocolIdNextHeader;

        v_FilterCount := v_FilterCount + 2

          }

      case (3){

        // Update to values in Reference packet filter #3

        v_Filter.iei         :=  '32'O; // Bidirectional filter, ID=1
        v_Filter.precedence  := '00001111'B;

      }

    }

    v_Filter.iel := int2oct(v_FilterCount, 1);

    return v_Filter;

  }


5 Execution Log Files

5.1 LG L2000 UE

The LG L2000 UE passed this test case on the Anite Conformance Toolset solution with Snow 3G integrity algorithm. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files tc_10_8_1_LG-log.html:

In the log files (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.

6 References

	[1]
	R5s100471: This archive comprises html format execution log file.

